
検索支援ツール「かんざし」 AHK版 v0.4 カスタマイズガイド 

2023年 8月 

この文書では、かんざし AHK版 v0.4の検索先や動作をカスタマイズする方法について説明します。 

デフォルト状態のかんざしの操作を試したうえでご覧ください。 

かなり長い文書ですが、すべてをお読みいただく必要はありません。目的に応じて、次のようにご利用ください。 

 検索先の追加やショートカットキーの変更などの基本的な修正については、「1 基本的なカスタマイズ」

で説明しています。カスタマイズを初めて行う方は、まずはこのセクションに一通り目を通すことをお勧

めします。 

 それ以外の細かなカスタマイズが必要な場合は、「2 目的別設定リファレンス」を参照して、目的に合

った箇所をお読みください。 

なお、かんざしの実行中に Kanzasi.ahk を修正したときには、タスクトレイでかんざしのアイコンを右クリックし、

「スクリプトの更新を反映」をクリックしてください。（これを行わないと修正が動作に反映されませんのでご注意

ください）。 

※スクリプトに手を加えるときには、修正を誤ると正しく動作しなくなる可能性があります。すぐに元に戻せるよ

う、Kanzasi.ahk のバックアップをとってから作業に入ることをお勧めします。 

1 基本的なカスタマイズ 

ここでは、デフォルト状態の Kanzasi.ahk に対する基本的なカスタマイズの方法として、以下の 6 つを説明し

ます。 

 かんざしの入力ウィンドウやポップアップメニューを表示するためのホットキーを変更する方法 

 検索先選択のショートカットキーや一発検索のホットキーを変更する方法 

 新たな検索先としてWebサイトを追加する方法 

 新たな検索先としてWindows上の辞書ソフトを追加する方法 

 Webサイトや辞書ソフトを串刺し検索する方法 

 独自の検索処理を記述する方法 

 

かんざしの入力ウィンドウやポップアップメニューを表示するためのホットキーを変更する 

かんざしの入力ウィンドウやポップアップを表示するためのホットキーは、Kanzasi.ahk の下記の部分で指定

できます。（「InputBoxActivate」で検索すると見つかります）。 

^+k::InputBoxActivate() ; 入力ウィンドウを表示 

^+d::InputBoxActivate(SelGet(false)) ; 選択語句を取り込んで入力ウィンドウを表示 

^+i::PopupMenuActivate(1, " ") ; 前回と同じ語句を検索するポップアップメニューをカーソル位置に表示 

^+e::PopupMenuActivate(1) ; 選択語句を検索するポップアップメニューをカーソル位置に表示 

MButton::PopupMenuActivate(0, , "{MButton}") ; 選択語句を検索するポップアップメニューをマウス位置に表示 

!^MButton::KanzasiInfoGet() ; スクリプト修正用の情報を取得して表示 

^+n::NextDictActivate(1) ; 次の辞書をアクティブ化 

^+p::NextDictActivate(-1) ; 前の辞書をアクティブ化 

キーを変更するには、上記の青色部分を修正してください。キーの指定方法は AHKの一般的な記述方法と同

じです。詳しくはこちらをご覧ください。 

検索先選択のショートカットキーや一発検索のホットキーを指定する 

かんざしの入力ウィンドウで各検索先を選択するためのショートカットキーや、ホットキー一発検索で使用する

ホットキーは、Kanzasi.ahk の下記の部分で指定できます。（「ini_grpid」で検索すると見つかります）。 

ini_grpid := {  3: "Google日本 = (^g), ((^+g))" 

              ,  6: "Google米国 = (^u), ((^+u))" 

              ,  9: "Google英国 = (^k)" 

              , 12: "Google米画像 = (^i)" 

              , 15: "Bing = (^b)" 

              , 18: "--------" 

              , 21: "EBWin4 = (^q), ((^+q))" 

              , 24: "--------" 

              , …（中略） 

              , 57: "Wiktionary = (^w)" 

              , 60: "M-W = (^s)" 

              , 63: "AHD = (^h)" 

              , 66: "--------" 

              , 69: "ロングマン英和 = (^;)" 

              , …（以下略） 

( ) 囲みの部分（青色マーカーの部分）がショートカットキー、(( )) 囲みの部分（緑色マーカーの部分）がホット

キーです。 

※ショートカットキーは ( ) 囲み、ホットキーは (( )) 囲みで指定します。ホットキーはWindows全体で幅広く有効な

キーだからカッコ (( )) も幅広い、と覚えるとよいかもしれません。 

このカッコ囲みの表記法は、かんざしで独自に定めたもので、この定義部分でのみ有効です。スクリプト内の他の部

分や、一般的な AHKスクリプトでは使えませんのでご注意ください。 

キーを別のものに変更するには、( ) 囲みや (( )) 囲みの中のキー指定を変更してください。キーの指定の仕

方は AHKの一般的な記述方法と同じです。詳しくはこちらをご覧ください。 

キーの追加や削除も可能です。たとえば、先ほどのコードで Bingの行を次のように変更すると、Ctrl + Shift + 

B というホットキーで Bing の一発検索が行えるようになります。 

             , 15: "Bing                 = (^b), ((^+b))" 

新たな検索先としてWebサイトを追加する 

新たな検索先を追加するための作業は、大きく 3つに分かれます。 

1. 検索先の名前、ショートカットキー、ホットキーを指定する 

2. 検索処理を記述するための「枠組み」を作成する 

3. 検索処理を記述する 

ここでは例として、辞書検索サイト「Onelook Dictionary Search」（https://onelook.com/）を新たな検索先とし

て追加する場合を想定して、この 3つの作業を説明していきます。 

1. 検索先の名前、ショートカットキー、ホットキーを指定する 

Kanzasi.ahk の中に次のような箇所があります。（ファイル内を「ini_grpid」で検索すると見つかります）。 

ini_grpid := {  3: "Google日本 = (^g), ((^+g))" 

              ,  6: "Google米国 = (^u), ((^+u))" 

              ,  9: "Google英国 = (^k)" 

              , 12: "Google米画像 = (^i)" 

              , 15: "Bing = (^b)" 

              , 18: "--------" 

              , 21: "EBWin4 = (^q), ((^+q))" 

              , 24: "--------" 

              , …（中略） 

              , 57: "Wiktionary = (^w)" 

              , 60: "M-W = (^m)" 

              , 63: "AHD = (^h)" 

              , 66: "--------" 

              , 69: "ロングマン英和 = (^;)" 

              , …（以下略） 

ここが、検索先の名前とキーを指定している部分です。1行が 1項目で、各行は 

「検索先 ID: "検索先名 = (ショートカットキー), ((ホットキー))"」という形式になっています。 

新たな検索先も同じ形式で追加します。この例では、次の青色部分のように記述してみましょう。 

ini_grpid := {  3: "Google日本 = (^g), ((^+g))" 

              ,  6: "Google米国 = (^u), ((^+u))" 

              ,  9: "Google英国 = (^k)" 

              , 12: "Google米画像 = (^i)" 

              , 15: "Bing = (^b)" 

              , 18: "--------" 

              , 21: "EBWin4 = (^q), ((^+q))" 

              , 24: "--------" 

              , …（中略） 

              , 57: "Wiktionary = (^w)" 

              , 60: "M-W = (^m)" 

              , 63: "AHD = (^h)" 

              , 64: "Onelook = (^1), ((^+1))" 

              , 66: "--------" 

              , …（以下略） 

https://onelook.com/

この行で記述した各項目の意味は次のとおりです。 

 「64」： 検索先の ID 番号。既存の番号と重複しない空き番号を指定してください。番号の並びが小さ

い順のままとなるような空き番号を選ぶことをお勧めします。また、ここで指定した値は後ほど使いま

すので、覚えておいてください。 

 「Onelook」： 検索先の名前。かんざしの入力ウィンドウやポップアップメニューでは、検索先としてこ

の名前が表示されます。 

 「(^1)」： ショートカットキー。かんざしの入力ウィンドウで、この検索先を選択するためのキーです。こ

の例で記述した ^1 とは「Ctrl + 1」という意味です。 

 「((^+1))」： 一発検索のホットキー。選択語句をホットキーのみで検索するときに使用するキーです。

この例で記述した ^+1 とは「Ctrl + Shift + 1」という意味です。 

※^1 や ^+1 のようなキーの指定方法は、AHK の標準の指定方法と同じです。詳しくはこちらをご覧ください。 

※検索先の名前と、その右のほうに出てくる「=」の間には、多数のスペースが入っていますが、このスペース

の数はいくつでもかまいません。（デフォルトでは、各行の「=」の位置が揃うようにスペースの数を調整していま

すが、別に揃っていなくても問題ありません）。 

※ショートカットキーやホットキーが不要の場合は、「, 64: "Onelook"」のように、ID番号と名前だけの指

定でもかまいません。 

2. 検索処理を記述するための「枠組み」を作成する 

Kanzasi.ahk の中に次のような箇所があります。（ファイル内を「DictGrp」で検索すると見つかります）。 

DictGrp3() ; Google日本 

{ 

    WebOpen(1, "https://www.google.co.jp/search?q=%s&hl=ja&gl=jp") 

} 

 

DictGrp6() ; Google米国 

{ 

    WebOpen(1, "https://www.google.co.jp/search?hl=en&gl=us&q=%s") 

} 

 

（中略） 

 

DictGrp60() ; M-W 

{ 

    WebOpen(1, "https://www.merriam-webster.com/dictionary/%s") 

} 

 

DictGrp63() ; AHD 

{ 

    WebOpen(1, "https://www.ahdictionary.com/word/search.html?q=%s") 

} 

 

DictGrp69() ; ロングマン英和 

{ 

    WebOpen(1, "https://www.ldoceonline.com/jp/dictionary/english-japanese/%s") 

} 

 

（以下略） 

この部分で、それぞれの検索先に対する検索処理を記述しています。 

個々の検索先に対する処理記述の枠組みは 

DictGrpXX() 

{ 

    ; 何らかの処理 

} 

 

という形式です。 

新しい検索先を追加するときも、まずはこの枠組みを作成します。このとき、「DictGrpXX」の「XX」の部分に

は、先ほど指定した検索先 ID 番号を記述してください。先ほどの例では「64」と指定しましたので、

「DictGrp64」という名前の枠組みを、次のように作成します。 

 

（前略） 

 

DictGrp60() ; M-W 

{ 

    WebOpen(1, "https://www.merriam-webster.com/dictionary/%s") 

} 

 

DictGrp63() ; AHD 

{ 

    WebOpen(1, "https://www.ahdictionary.com/word/search.html?q=%s") 

} 

 

DictGrp64() 

{ 

 

} 

 

DictGrp69() ; ロングマン英和 

{ 

    WebOpen(1, "https://www.ldoceonline.com/jp/dictionary/english-japanese/%s") 

} 

 

（以下略） 

3. Webサイトに対する検索処理を記述する 

上で作成した 「DictGrpXX() { ... }」の枠組みの中に、Webサイトの検索処理本体を記述していきま

す。 

Webサイトの検索処理は、上の例にある他サイトと同様に、 

DictGrp64() 

{ 

    WebOpen(1, "https://～") 

} 

という形式で、URLを埋め込む形で記述します。 

URLの部分は、検索対象のサイトをWebブラウザで開き、手動で検索して、その結果ページから取得してくだ

さい。今回の例なら、目的の Onelook Dictionary Search （https://onelook.com/）を開いて、たとえば

「example」という単語を検索してみると、検索結果ページの URLは次のようになっています。 

https://onelook.com/?w=example&ls=a 

この検索結果の URL をコピーして、先ほどのコードで青色になっていた "https://～" という部分に貼り付けた

うえで、検索語句の箇所（上の例なら緑色の「example」の箇所）を「%s」に置き換えてください。 

※検索結果のURLに検索語句が含まれていない場合、この形式のコードでは検索できません。対処方法はこちらを

ご覧ください。 

つまり今回の例では、Onelookを検索する処理は、下記の青色部分のように記述することになります。 

DictGrp64() 

{ 

    WebOpen(1, "https://onelook.com/?w=%s&ls=a") 

} 

以下で検索先の登録は完了です。かんざしの入力ウィンドウやポップアップメニューを開いて、実際に検索でき

るかどうか確認してみてください。 

新たな検索先としてWindows上の辞書ソフトを追加する 

Windows上の辞書ソフトを新たな検索先として追加する作業も、全体は大きく 3つに分かれます。 

1. 検索先の名前、ショートカットキー、ホットキーを指定する 

2. 検索処理を記述するための「枠組み」を作成する 

3. 検索処理を記述する 

このうち 1. と 2. は、前項で説明した手順とまったく同じです。ここでは、3. の検索処理の記述方法のみ説

明します。 

検索処理の枠組みを次のように作ったとします。 

DictGrp22() 

{ 

     

} 

この中に記述する、辞書ソフトの検索処理のコードは、以下の手順で取得できます。 

https://onelook.com/

(1) Kanzasi.ahk をダブルクリックして、AHK版かんざしを起動する。 

(2) 検索対象の辞書ソフトを起動する。 

(3) 辞書ソフトの語句入力エリアの上で、Ctrlキーと Alt キーを押しながらマウスの中ボタンをクリックする。（マ

ウスのホイール部分をクリックして押し下げてください。ホイールの回転とは異なります）。 

すると、かんざしの入力ウィンドウがポップアップし、クリック先の辞書ソフトを検索するためのコードが次の

ような形で表示されます。 

(4) 表示された内容全体をクリップボードにコピーする（Ctrl+Aで全選択し、Ctrl+Cでコピー）。 

(5) 冒頭で示した DictGrpXX() { ... } の枠組みの中に、その内容を貼り付ける。 

例：（実際のコードは検索先のソフトによって異なります） 

DictGrp22() 

{ 

    WinClickSend("ahk_class TLogophileMainForm", "TEdit2", "{Home}+{End}^v{Enter}") ; 「Oxford Learner's 

THESAURUS」, "x332 y69" 

} 

これで検索先の登録は完了です。実際に検索できるかどうか確認してみてください。（検索先の辞書ソフトを起

動した状態でお試しください） 

※この方法でうまく検索できないときには、貼り付けたコードに修正を加えることで検索できるようになる場合がありま

す。詳しくはこちらをご覧ください。 

※上記で貼り付けたコードで、カッコ内の 2 番目の項目が "x○○ y○○" という形式の場合、辞書ソフトのウィンド

ウ内で入力エリアの位置が変わると、検索時に予期せぬ動作を起こす可能性が高くなります。ウィンドウのサイズや

表示項目の配置を不用意に変更しないようご注意ください。 

WinClickSend("ahk_class TMainForm", "x158 y624", "{Home}+{End}^v{Enter}") ; ... 

 ↑ここが座標指定の場合は注意！ 

Webサイトや辞書ソフトを串刺し検索する 

複数のソフトやサイトを 1 回の操作で串刺し検索するには、DictGrpXX() の枠組みの中に、目的のソフトやサ

イトに対する検索処理を並べて記述しておけば OKです。たとえば次のように記述すると、Google と Bingを 1

回の操作で串刺し検索できます。 

DictGrp16() 

{ 

    WebOpen(1, "https://www.google.co.jp/search?q=%s&hl=ja&gl=jp") 

    WebOpen(2, "https://www.bing.com/search?q=%s") 

} 

※上の例で、WebOpen の直後の「1」「2」という数字は、検索結果を Web ブラウザのどのタブで開くかを表します。

上の例なら、Googleの検索結果は先頭のタブで、Bingの結果は 2番目のタブで、それぞれ開きます。 

Webサイトに対する検索と、Windows上の辞書ソフトに対する検索が混在していても問題ありません。 

独自の検索処理を記述する 

ここまでの例では、DictGrpXX() の枠組みの中で、WebOpen(...) や WinClickSend(...) など、かんざしが用

意しているヘルパー関数を使って検索処理を記述してきました。このほか、DictGrpXX() の中では、AHKの一

般的なコードと同じような形で、独自の検索処理のロジックを記述していただいてもかまいません。検索語句は

クリップボードに入っていますので、Ctrl+V で検索語句の入力フィールドに貼り付けたり、Clipboard 変数を使

って操作したりできます。 

デフォルトのコードでは、EBwin を検索する DictGrp21() がそのような処理の例です。EBwinの exeファイル

に検索語句を起動パラメータとして渡すという方法で、検索処理を実行しています。 

DictGrp21() ; EBWin 

{ 

    ebwin64 := "C:¥Program Files¥EBWin4(x64)¥EBWin4.exe" 

    ebwin32 := "C:¥Program Files (x86)¥EBWin4¥EBWin4.exe" 

     

    ebwinexe := FileExist(ebwin64) ? ebwin64 : FileExist(ebwin32) ? ebwin32 : "" 

    if (ebwinexe != "") { 

        Run, % ebwinexe . " /S=""" . StrRepl(Clipboard, """", "") . """" 

    } 

} 

また、たとえば次のようなコードを記述すると、Google Chrome のウィンドウをアクティブ化して、Ctrl + F で検

索ボックスを開き、ページ内検索を実行できます。 

DictGrp100() 

{ 

    if (WinExist("Google Chrome")) { 

        WinActivate 

        Sleep, 100 

        Send, ^f 

        Sleep, 100 

        Send, ^v 

    } 

} 

2 目的別 設定リファレンス 

主な設定項目について、修正箇所を目的別にまとめました。 

目的 設定項目 

入力ウィンドウのデフォルトの位置とサイズを変更

するには 
ini_x, ini_y, ini_width を修正 

「入力ウィンドウを常に表示」モードをデフォルトで

オンにするには 
ini_stayopen を修正 

辞書ソフトやブラウザが未起動の場合に自動で起

動してから検索させるには 
ini_winexe を設定 

検索先を新規作成・修正・削除するには 
ini_grpid で ID番号・検索先名・ショートカットキー等

を定義し、DictGrpXX() で検索処理を記述 

検索先の名前を変更するには ini_grpid でグループ名の定義を修正 

入力ウィンドウ上で辞書グループを選択するため

のショートカットキーを変更するには 
ini_grpid で ( ) 囲みのキー定義を修正 

ホットキー一発検索のキーを変更するには ini_grpid で (( )) 囲みのキー定義を修正 

Windows 上の辞書ソフトを検索先として追加する

には 
DictGrpXX() に追加 

Windows 上の辞書ソフトが既製のコードでうまく検

索できない場合は 
DictGrpXX() 内の WinClickSend() の記述を修正 

URLに検索語句が含まれるタイプのWebサイトを

検索先として追加するには 
DictGrpXX() に追加 

URLに検索語句が含まれないタイプのWebサイト

を検索先として追加するには 
DictGrpXX() に追加 

検索結果を Web ブラウザで開くとき、どのタブで開

くかを指定するには 

ini_browserwin でブラウザを指定したうえで、

DictGrpXX() 内の WebOpen() または 

WebOpenSend() にタブ番号を指定 

入力ウィンドウやポップアップメニューを表示するホ

ットキーを変更するには 
ホットキー部のキー定義を変更 

入力ウィンドウで使用する短縮入力を指定するに

は 

Kanzasi_hotstring.txt を開いて編集するか、ショー

トカット部またはホットキー部で定義 

 

次のページからは、スクリプト内の個別の構成要素の方法について説明していきます。 

細かい話が延々と続きますので、上の表を参照のうえ、必要になったときに必要な箇所のみお読みいただくこと

をお勧めします。 

3 スクリプトの内容 

スクリプト全体の構成 

スクリプト全体は次のような構成となっています。設定ウィンドウでいうタブページのようなイメージで、設定内

容ごとにセクションが分かれていると捉えていただくとよいと思います。 

 
; AutoHotkey版「かんざし」Version 0.4 
 
; このファイルをダブルクリックすると処理が起動します。 
; このファイルは自由にカスタマイズしてご利用ください。 
; Kanzasi_lib.ahk のほうは手を加えずお使いいただくことをお勧めします。 
 
#Include %A_ScriptDir%\Kanzasi_lib.ahk 
 
Kanzasi_Initialize: 
 
SendMode Event                                              ; Sendでキーを送るときの入力モード（通常は Inputか Eventのどちらか） 
SetKeyDelay, 0                                              ; Eventモードで Sendするときのキー間隔（※Input モードでは適用されません） 
 
;■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■ 
;■■■■■■■■■■■■■■ ＧＵＩの設定 ■■■■■■■■■■■■■■■ 
;■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■ 
 
ini_x := -100, ini_y := 0, ini_width := 300                 ; 入力ウィンドウの位置と幅（ini_xを-100にすると起動時は画面中央に表示、-200にすると常にアクティブウィンドウ中央に表示） 
 
ini_stayopen       := false                                 ; 入力ウィンドウを常に表示しておくかどうか（true/false） 
ini_stayontop      := true                                  ; 入力ウィンドウを表示する時、常に前面表示とするかどうか（true/false） 
ini_fontname       := "ＭＳ Ｐゴシック"                     ; 入力ウィンドウのフォント名 
ini_fontsize       := 13                                    ; 入力ウィンドウのフォントの大きさ 
ini_activate1stwin := true                                  ; 語句送信後に辞書ウィンドウをアクティブ化するかどうか（true/false） 
ini_logfilename    := ""                                    ; ログファイル名。空なら記録しない。右辺は "C:\～" のような形式でフルパスで書いても可 
 
;■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■ 
;■■■■■■■■■■ ブラウザ・自動起動ソフトの設定 ■■■■■■■■■■ 
;■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■ 
 
ini_browserwin := ""                                        ; Web検索で使用するブラウザ 
 
ini_winexe := { "" : ""                                     ; ウィンドウクラス（and/orタイトル）と EXEファイルの対応付け（先頭行はダミー） 
    , "この行にコードを貼り付け" : "" } 
 
 
;■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■ 
;■■■■■■■■■■ 検索先の ID番号・名前・キーの設定■■■■■■■■■■ 
;■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■ 
 
                                                            ; 辞書グループの番号・名前・ショートカットキー・ホットキーの定義 
ini_grpid := {  3: "Google日本           = (^g), ((^+g))" 
             ,  6: "Google米国           = (^u), ((^+u))" 
             ,  9: "Google英国           = (^k)" 
             , 12: "Google米画像         = (^i)" 
             , 15: "Bing                 = (^b)" 
             , 18: "--------" 
             , 21: "EBWin                = (^q), ((^+q))" 
             , 24: "--------" 
             , 27: "学習英英串刺し       = (^5)" 
             , 30: "一般英英串刺し       = (^6)" 
             , 33: "日本語類語串刺し     = (^7)" 
             , 36: "--------" 

（中略） 
             , 90: "--------" 
             , 93: "Weblio類語           = (^8)" 
             , 96: "翻訳類語辞典         = (^9)" 
             , 99: "類語玉手箱           = (^0)" 
             ,102: "連想類語辞典         = (^-)"} 
 
 
KanzasiCommonInit()                                         ; 共通の初期処理を実行（※この行は必須です。削除しないようご注意ください） 
 
return                                                      ; 初期処理終了 
 
 
;■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■ 
;■■■■■■■■■■ 各検索先に対する検索処理の設定 ■■■■■■■■■■ 
;■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■ 
 
DictGrp3() ; Google日本 
{ 
    WebOpen(1, "https://www.google.co.jp/search?q=%s&hl=ja&gl=jp") 
} 
 
DictGrp6() ; Google米国 
{ 
    WebOpen(1, "https://www.google.co.jp/search?hl=en&gl=us&q=%s") 
} 
 
DictGrp9() ; Google英国 
{ 
    WebOpen(1, "https://www.google.co.jp/search?hl=en&gl=uk&q=%s") 
} 
 
DictGrp12() ; Google米画像 
{ 
    WebOpen(1, "https://www.google.co.jp/search?hl=en&gl=us&tbm=isch&q=%s") 
} 
 

（中略） 
 
DictGrp99() ; 類語玉手箱 
{ 
    WebOpen(1, "https://www.thesaurus-tamatebako.jp/?s=%s") 
} 
 
DictGrp102() ; 連想類語辞典 
{ 
    WebOpen(1, "https://renso-ruigo.com/word/%s") 
} 
 

（続く） 

GUIの設定 

Webブラウザの設定・自動起動するソ

フトの設定 

検索先の ID番号・名前・ショートカット

キー・ホットキーの設定 

各検索先に対する検索処理の設定 

;■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■ 
;■■■■■■■■■■■■■ ホットキーの設定 ■■■■■■■■■■■■■■ 
;■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■ 
 
^+k::InputBoxActivate()                                     ; 入力ウィンドウを表示 
^+d::InputBoxActivate(SelGet(false))                        ; 選択語句を取り込んで入力ウィンドウを表示 
^+i::PopupMenuActivate(1, " ")                              ; 選択語句を検索するポップアップメニューをカーソル位置に表示 
^+e::PopupMenuActivate(1)                                   ; 選択語句を検索するポップアップメニューをカーソル位置に表示 
MButton::PopupMenuActivate(0, , "{MButton}")                ; 選択語句を検索するポップアップメニューをマウス位置に表示 
!^MButton::KanzasiInfoGet()                                 ; スクリプト修正用の情報を取得して表示 
^+n::NextDictActivate(1)                                    ; 次の辞書をアクティブ化 
^+p::NextDictActivate(-1)                                   ; 前の辞書をアクティブ化 
 
 
;■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■ 
;■■■■■■■■■■■ ショートカットキーの設定 ■■■■■■■■■■■■ 
;■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■ 
 
#IfWinActive かんざし - ahk_class AutoHotkeyGUI 
 
^2:: 
    hk_text := InputBoxGetText() 
    hk_text := InStr(hk_text, """") ? StrRepl(hk_text, """") : """" . hk_text . """" 
    InputBoxSetText(hk_text, 2) 
    return 
 
^3:: 
    hk_text := InputBoxGetText() 
    hk_text := InStr(hk_text, """") ? StrRepl(hk_text, """") : RegExReplace(hk_text, "(*UCP)(\S+)", """$1""") 
    InputBoxSetText(hk_text, 2) 
    return 
 
#IfWinActive 
 
 
;■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■ 
;■■■■■■■■■■■■■■ 検索の前処理 ■■■■■■■■■■■■■■■ 
;■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■■ 
 
DictGrpPreprocess(grp) 
{ 
 
} 
 
WebOpenPreprocess(queryurl) 
{ 
 
} 
 

  

入力ウィンドウやポップアップメニュー

を表示するホットキーの設定 

入力ウィンドウ上のショートカットキー

の設定 

高度なカスタマイズのための枠組み 

（ひとまず無視して大丈夫です） 

GUIの設定 

入力ウィンドウの表示などに関する設定を行うセクションです。 

デフォルト設定 

ini_x := -100, ini_y := 0, ini_width := 300 

 

ini_stayopen       := false 

ini_stayontop      := true 

ini_fontname       := "ＭＳ Ｐゴシック" 

ini_fontsize       := 13 

 

ini_activate1stwin := true 

設定項目 

入力ウィンドウのデフォルトの位置とサイズ 

項目 ini_x, ini_y, ini_width 

指定値 入力ウィンドウの上で「Ctrl+Alt+中ボタン」し、表示されたコードをそのまま上書きで貼り付け 

説明 実行中に入力ウィンドウの位置とサイズを変更しても、自動では記憶されず、再起動すると

デフォルトの状態に戻ります。その「デフォルトの状態」を指定するための場所がここです。 

入力ウィンドウを目的の位置とサイズに調整したうえで「Ctrl+Alt+中ボタン」し、表示された内

容を貼り付けてください。 

なお、位置に関しては、特別な設定値として次の 2つがあります。 

 ・ini_x := -100 …入力ウィンドウを画面中央に表示 

 ・ini_x := -200 …入力をアクティブウィンドウの中央に表示 

（これらの値を指定するときは、ini_yの値は無視されます）。 

「入力ウィンドウを常に表示」をデフォルトでオンにするかどうか 

項目 ini_stayopen 

指定値 true / false 

説明 実行中にタスクトレイで設定を変更できる項目ですが、それを起動時点でオンにするかどう

かをここで指定します。 

入力ウィンドウを常に前面表示するかどうか 

項目 ini_stayontop 

指定値 true / false 

入力ウィンドウのフォントの種類 

項目 ini_fontname 

指定値 フォント名を " " で囲んで指定 

入力ウィンドウのフォントのサイズ 

項目 ini_fontsize 

指定値 フォントサイズを表す数値 

送信完了後に先頭の辞書をアクティブ化するかどうか 

項目 ini_activate1stwin 

指定値 true / false 

説明 falseの場合は、送信前にアクティブだったウィンドウに戻ります。 

（ただし作者の環境では、入力ウィンドウが閉じた直後になぜかフォーカスが別のウィンドウ

に移ってしまうことがあり、false の場合には予期したウィンドウとは異なるウィンドウに戻って

しまうことがあります）。 

Webブラウザの設定・自動起動するソフトの設定 

Web検索に使うブラウザと、未起動時に自動起動する辞書ソフト類を指定するセクションです。 

設定例 

ini_browserwin := "ahk_class MozillaWindowClass" 

 

ini_logfilename    := "" 

 

ini_winexe := { "" : "" 

    , "ahk_class TJammingMainForm" : "C:¥Program Files (x86)¥Jamming¥Jamming.exe,1000" } 

設定項目 

Web検索で使用するブラウザ 

項目 ini_browserwin 

指定値 Google Chromeの場合は"chrome"、Mozilla Firefoxの場合は"firefox"、Microsoft Edgeの

場合は"edge"。それ以外のブラウザの場合は、ブラウザのウィンドウ上で「Ctrl+Alt+中ボタ

ン」し、ドロップダウンの「〈ウィンドウクラス〉」の内容を " " で囲んで貼り付け 

説明 デフォルトでは、Web検索の結果は、Windows上で「既定のブラウザ」として設定しているブ

ラウザの新しいタブで表示されますが、ここでブラウザを明示的に指定しておくことをお勧め

します。（別の場所の設定と併用することにより、検索結果をどのタブで開くかを指定できる

ようになるからです）。 

ログファイル名 

項目 ini_logfilename 

指定値 ログファイル名を表す文字列 

説明 ここにファイル名を指定すると、検索を実行するたびにそのファイルに履歴が記録されるよう

になります。ファイルが存在しない場合は自動で作成されます。 

※古い履歴を自動で削除する機能はありません。長い間使用していると、ログファイルがど

んどん大きくなっていきます。お手数ですが、必要に応じてご自身で整理してください。（ログ

ファイルをテキストエディタで開いて、不要な行を手動で削除すれば大丈夫です）。 

 

  



 

16 

 

自動起動するソフトの指定 

項目 ini_winexe 

指定値 対象の辞書ソフトやブラウザのウィンドウ上で「Ctrl+Alt+中ボタン」し、ドロップダウンの

「〈EXE自動起動用コード〉」の内容を貼り付けて、行末に「}」を追加 （この後の説明を参照） 

説明 語句を送信しようとした時点で辞書ソフトやブラウザが未起動だった場合に、自動で起動さ

せたい場合は、そのソフトをここで指定しておく必要があります。 

ini_winexe は「Ctrl+Alt+中ボタン」からのコピペで指定できます。「〈EXE 自動起動用コード〉」の内容を、下記

の青色の部分に貼り付けてください。 

ini_winexe := { "" : ""                  ; ウィンドウクラス（and/or タイトル）と EXE ファイルの対応付け（先頭行はダミー） 

    , "この行にコードを貼り付け" : "" } 

自動起動させたいソフトが複数ある場合は、それぞれに対する「Ctrl+Alt+中ボタン」の「〈EXE 自動起動用コー

ド〉」を、1行に 1つずつ、並べて貼り付けてください。 

そのうえで、最後の行の末尾にのみ「}」を加えて、次のような形としてください。 

 

 

 

貼り付けたコードの exeのフルパスが間違っていないかどうか、念のためご確認ください。 

また、exe のパスの末尾に「,1000」とあるのは、その exe を起動した後で何ミリ秒待ってから検索語句を送る

かを表しています。起動の後の初期処理に時間がかかってすぐには検索できないソフトの場合は、ここの数字

を大きくしてください。 

  

ini_winexelist := { "" : ""                  ; ウィンドウクラス（and/or タイトル）と EXE ファイルの対応付け（先頭行はダミー） 

  , "ahk_class TJammingMainForm" : "C:¥Program Files (x86)¥Jamming¥Jamming.exe,1000" 

  , "ahk_class Chrome_WidgetWin_1" : "C:¥Program Files (x86)¥Google¥Chrome¥Application¥chrome.exe,1000" } 

最終行のみ、末尾に中カッコ閉

じを付ける 各行の先頭はカンマ 

検索先の ID番号・名前・ショートカットキー・ホットキーの設定 

検索先の ID番号・名前・ショートカットキー・ホットキーを設定するセクションです。 

デフォルト設定 

ini_grpid := {  3: "Google 日本           = (^g), ((^+g))" 

             ,  6: "Google 米国           = (^u), ((^+u))" 

             ,  9: "Google 英国           = (^k)" 

             , 12: "Google 米画像         = (^i)" 

             , 15: "Bing                 = (^b)" 

             , 18: "--------" 

             , 21: "EBWin4               = (^q), ((^+q))" 

             , 24: "--------" 

             , …（以下略） 

設定項目 

1行が 1項目で、「検索先 ID番号: "検索先名 = (ショートカットキー), ((ホットキー))"」という形式です。 

 既存の検索先の名前やキーを修正する場合は、該当する箇所を直接書き換えてください。 

 新しい検索先を作成する場合は、デフォルトの記述をコピペする形で行を追加し、各項目を書き換えて

ください。検索先はいくつでも作成できます。 

 検索先を削除する場合は、行ごと削除してください。 

それぞれの項目の内容と指定値は次のとおりです。 

検索先 ID番号 

内容 スクリプト内で検索先を区別するために付ける番号 

指定値 正の整数値 

説明 

他と重複しない値を自由に付けてかまいません。ポップアップメニューの検索先名はこの番

号順に表示されます。 

ここで指定した番号は、実際の検索処理を記述する箇所でも使います。 

検索先名 

内容 入力ウィンドウやポップアップメニューで表示される検索先名 

指定値 任意の文字列 

ショートカットキー 

内容 入力ウィンドウ上で検索先を選択するためのショートカットキー 

指定値 キーを表す文字列を ( ) ［半角カッコ 1つの組］で囲んで指定。 

カンマ区切りでいくつでも指定可能 

説明 
キーの文字列は AHKの書式どおりです。「ホットキー設定ヘルパー」を使うと簡単です。 

ショートカットキーが不要な検索先は、指定を省略してかまいません。 

ホットキー 

内容 検索先を一発検索するためのホットキー 

指定値 
キーを表す文字列を (( )) ［半角カッコ 2つの組］で囲んで指定。 

カンマ区切りでいくつでも指定可能 

説明 
キーの文字列は AHKの書式どおりです。「ホットキー設定ヘルパー」を使うと簡単です。 

ホットキー一発検索が不要な検索先は、指定を省略してかまいません。 

※ショートカットキーは ( ) で囲み、ホットキーは (( )) で囲みます。ホットキーは幅広いウィンドウで有効なキーだか

らカッコ (( )) も幅広い、と覚えるとよいかもしれません。 

なお、このカッコ囲みの表記法は、かんざしで独自に定めたものであり、この部分でのみ有効です。スクリプト内の他

の部分や、一般的な AHKスクリプトでは使えませんのでご注意ください。 

記述方法に関する注意・補足 

 "検索先名 = (ショートカットキー), ((ホットキー))" という部分は、この部分の文字列全体を " " で囲ん

でください。（左辺と右辺の各項目を " " で囲む形ではありません）。 

 ショートカットキーとホットキーの組み合わせは、すべての検索先を通して一意でなくてはなりません。

重複がないように指定してください。 

 ショートカットキーもホットキーも不要な検索先は、たとえば「15: "Bing"」のように、検索先名のみを " " 

囲みで指定する形でかまいません。 

 ポップアップメニューで区切り線を入れたい箇所には、検索先名に "--------"（半角のマイナス 8個）

を指定してください。その項目にも ID番号は必要ですが、DictGrpXXは不要です。 

 定義の修正が終わったら、次のような形式になっているかどうかご確認ください。 

 

 

 

 

なお、この ini_grpid で検索先を追加した場合は、この後の説明に従って、DictGrpXX の検索先も指定してく

ださい。（区切り線の "--------" の項目には DictGrpXXは必要ありません）。 

ini_grpid := { 10: "ローカル辞書  = (^d),(F1),((^+d))" 

            , 20: "Google日本  = (^g),(F2),((^+g))" 

            , 30: "Google US   = (^u),(F3)" 

            , 40: "オンライン英英 = (^e),(F4)" 

            , 50: "オンライン類語 = (^r),(F5)" } 

 

先頭に中カッコ開きがある 

最終行のみ、末尾に中カッコ閉

じがある 

2 行目以降の各行の先頭にカン

マがある 

各検索先に対する検索処理の設定 

それぞれの検索先について、辞書ソフトやWebサイトに対する検索処理を記述するセクションです。 

DictGrp3() ; Google日本 

{ 

    WebOpen(1, "https://www.google.co.jp/search?q=%s&hl=ja&gl=jp") 

} 

 

DictGrp6() ; Google米国 

{ 

    WebOpen(1, "https://www.google.co.jp/search?hl=en&gl=us&q=%s") 

} 

 

DictGrp9() ; Google英国 

{ 

    WebOpen(1, "https://www.google.co.jp/search?hl=en&gl=uk&q=%s") 

} 

（以下略） 

前述の ini_grpid で指定した検索先 ID 番号ごとに、DictGrpXX() { ... } という枠組みを用意し、それぞれの 

{ } 囲みの中に検索先を記述していきます。複数の検索先を 1回の操作で検索させたい場合は、{ } 囲みの中

に複数の検索処理を、たとえば下記のように記述していきます（Windows上の辞書ソフトに対する検索とWeb

検索が混在していても問題ありません）。 

DictGrp60()  

{ 

    WinClickSend(....) ; ローカル検索 

    WinClickSend(....) ; ローカル検索 

    WebOpen(0, "http://www.google.com/search?hl=en&q=%s") ; Web検索 

    WebOpen(0, "http://www.onelook.com/?w=%s&ls=a") ; Web検索 

} 

検索先の種類 

検索先は、主に次の 3種類に分かれます。 

 ローカル辞書ソフト：Windows 上の辞書ソフトやアプリケーションの入力エリアに語句を送信する検索

です。（設定方法はこちら） 

 URL型検索サイト：Web 上の検索サイトや辞書サイトのうち、検索結果の URLの中に検索語句が含

まれているタイプのサイトに対する検索です。（設定方法はこちら） 

 非 URL型検索サイト：Web上の検索サイトや辞書サイトのうち、検索結果の URLの中に検索語句が

含まれていないタイプのサイトに対する検索です。できれば、該当の検索ページを開いた時点で、検索

語句の入力フィールドにカーソルが置かれるページのほうが好ましいです。（設定方法はこちら） 

※スクリプトにデフォルトで登録してあるWeb検索先は、いずれも URL型の検索サイトです。 

※非 URL型サイトへの検索処理は、多少ぎくしゃくした動きとなり、少し時間がかかります。URL型と非 URL型のど

ちらの方法でも検索できるサイトは、URL型として検索することをお勧めします。 

以下、それぞれの設定方法を説明します。 

ローカル辞書ソフトを検索先として指定する方法 

ローカル辞書ソフトを検索先に追加する場合は、検索対象のソフトの入力エリア上で「Ctrl+Alt+中ボタン」し、

表示されたコードを、DictGrpXX() のブロックの中に貼り付けてください。 

貼り付けたコードで正しく検索できない場合の対処法 

辞書ソフトによっては、「Ctrl+Alt+中ボタン」から貼り付けたコードで正しく検索できない場合がありますが、次

に示す修正方法により検索できる可能性があります。 

○ インクリメンタルサーチの辞書ソフトの場合 

キー入力と並行して検索が実行される「インクリメンタルサーチ」型の辞書ソフトの場合は、最後に送信する

Enterキーが不要となります。貼り付けたコードで、カッコ内の末尾にある項目から{Enter}を削除してください。 

    WinClickSend(..., "{Home}+{End}^v{Enter}") 

 ↑この部分を削除 

○ 前の検索語句が残ってしまう場合 

前の検索語句が消えずに次の語句が送られている場合は、貼り付けたコードでカッコ内の末尾にある

{Home}+{End}^v{Enter} という部分を修正すると直る可能性があります。 

    WinClickSend(..., "{Home}+{End}^v{Enter}") 

 ↑この部分を修正 

 入力エリアを Ctrl+Aで全選択できる辞書ソフトの場合は、"^a^v{Enter}" に変えてみてください。 

 入力エリアの右クリックメニューに「すべて選択 (A)」という項目がある辞書ソフトの場合は、

"+{F10}^v{Enter}" に変えてみてください。 

 ウィンドウがアクティブ化した時点で入力エリアが全選択状態になる辞書ソフトの場合は、"^v{Enter}" 

に変えてみてください。 

○ ウィンドウ自体がアクティブにならない場合 

ウィンドウ自体がアクティブにならない場合は、貼り付けたコードのカッコ内の先頭項目をウィンドウのタイトル

に変更してみてください。（タイトルの一部のみでもかまいません）。 

タイトルの内容は、貼り付けたコードの末尾のほうに 「 」 囲みで記載してあります。 

    WinClickSend("ahk_class olt1UIWindowClass", ... ) ; 「～」 

 ↑この部分をウィンドウのタイトルに変更 ↑タイトルはここにあり 

 （実際の内容はプログラムにより異なります） 

○ ウィンドウはアクティブになるがキーが送られない場合 

貼り付けたコードでカッコ内の 2番目にある項目を、入力エリアの座標による指定に変更すると、検索できる可

能性があります。 

座標は、貼り付けたコードの末尾に "x○○○ y○○○" のような形式で記述してあります。その形式のままコ

ピペしてください。 

    WinClickSend("ahk_class ～", "olt1WindowClass13" ... ) ; 「～」, "x130 y112" 

 ↑この 2番目の項目を座標に変更 ↑座標はここにあり 

 （実際の内容はプログラムにより異なります） 

※このように座標を直接指定する形のコードにした場合、辞書ソフトのウィンドウ内で入力エリアの位置が変わったと

きに予期せぬ動作となる可能性がありますのでご注意ください。 

○ 登録した直後は検索できたが、辞書ソフトを再起動したら検索できなくなった場合 

辞書ソフトによっては、送信先のウィンドウと入力エリアを特定するための情報が、起動のたびに変わってしま

うものがあります。その場合にこの現象が発生します。 

上で説明した「ウィンドウ自体がアクティブにならない場合」「ウィンドウはアクティブになるがキーが送られない

場合」の内容にならって、貼り付けたコードのカッコ内の 1 番目の項目をウィンドウのタイトルに、2 番目を入力

エリアの座標に、それぞれ変更してみてください。（どちらか一方の修正のみで済む場合もあります）。 

URL型検索サイトを検索先として指定する方法 

検索結果の URL に検索語句が含まれている「URL 型」の検索先を追加するときには、DictGrpXX() のブロッ

クの中に、「WebOpen(0, "http～")」という形式でコードを記述します。 

URL の部分は、検索語が入る位置を「%s」で置き換えた形で指定します。目的の Web サイトで、たとえば

「sample」といった適当な単語を検索し、検索結果ページの URL をコピペしてから、「sample」とある部分を

「%s」に置き換えれば、たいていは動作すると思います。 

%s の部分には、文字コード「UTF-8」で検索語句が挿入されます。もし文字化けが生じる場合、URL に文字コードの

設定オプションがあれば、UTF-8 となるよう設定してみてください。 

非 URL型検索サイトを検索先として指定する方法 

検索結果の URLに検索語句が含まれていない「非 URL型」の検索サイトも、かんざしから検索可能です。 

できれば、該当の検索ページを開いた時点で、検索語句の入力フィールドにカーソルが置かれるタイプのペー

ジのほうが好ましいです。 

○ 非 URL型で、検索ページを開いたときに入力フィールドにカーソルが置かれるサイトの場合 

Webページ内の検索語句入力フィールドの上で「Ctrl+Alt+中ボタン」し、ドロップダウンの「〈非URL型Web検

索用コード〉」を DictGrpXX() の中に貼り付けたうえで、URL を指定してください。URL は「%s」のないものを

そのまま記述してください。 

なお、非 URL 型検索サイトの検索は、処理の都合上、多少ぎくしゃくして時間がかかる形の実装となっており

ます。ご了承ください。 

○ 非 URL型で、検索ページを開いたときに入力フィールドにカーソルが置かれないサイトの場合 

こうした形式のサイトも一応は検索可能です。「Ctrl+Alt+中ボタン」の「〈非 URL 型Web 検索用コード〉」からコ

ードを貼り付けたうえで、カッコ内の 5 番目の項目を、末尾にある "x○○ y○○" という座標に変えてください。 

例： 

WebOpenSend(0, "http://kod.kenkyusha.co.jp/demo/form.jsp", 10, "研究社辞書検索", "", "^a^v{Enter}") ; "x796 y259" 

 ↑この青色の部分を、 

 行末にある緑色の内容に変更 

※ただし、このように座標を直接指定する形のコードにした場合、ページのサイズや配置が変わったときに予期せぬ

動作となる可能性がありますのでご注意ください。 

Webページを開くタブ番号の指定 

デフォルトでは、Web 検索の結果は、Windows 上で「既定のブラウザ」として設定しているブラウザの新しいタ

ブで表示されますが、Google Chrome、Mozilla Firefox、Microsoft Edgeの場合は、検索結果を開くタブを指

定できます。たとえば、Google の検索結果は常にブラウザの 1 番左のタブで開く、といったように、何をどのタ

ブを開くか固定しておくと便利です。 

この機能を使うためには、次の 2つの両方を行う必要があります。 

 検索に使うブラウザの設定。 

「ブラウザの設定」セクションで説明した「ini_browserwin」という項目で設定してください。 

 検索を実行するコードでのタブ番号の指定。 

検索用コードのカッコ内の先頭項目（下記の青色の部分）に、目的のタブ番号を表す数字を指定してく

ださい。 

・ URL型検索サイトの場合： WebOpen(0, "http://～") 

・ 非 URL型検索サイトの場合： WebOpenSend(0, "http://～", ...) 

タブ番号として指定できる数字と動作は次のとおりです。 

タブ番号の指定 動作 

0 新しいタブで開く 

1～8 左から n番目のタブで開く（1=一番左のタブ、2=左から 2番目のタブ、…） 

9 一番右のタブで開く 

その他 現在のタブで開く 

※1～9を指定した場合、該当するタブがないときにどうなるかはブラウザ次第です。なるべく、常に存在するタブの番

号を指定してください。 

入力ウィンドウやポップアップメニューを表示するホットキーの設定 

入力ウィンドウやポップアップメニューを表示するホットキーを設定するセクションです。 

※一発検索のホットキーは、「検索先の ID 番号・名前・ショートカットキー・ホットキーの設定」セクションで定義します。 

^+k::InputBoxActivate() ; 入力ウィンドウを表示 

^+d::InputBoxActivate(SelGet(false)) ; 選択語句を取り込んで入力ウィンドウを表示 

^+i::PopupMenuActivate(1, " ") ; 前回と同じ語句を検索するポップアップメニューをカーソル位置に表示 

^+e::PopupMenuActivate(1) ; 選択語句を検索するポップアップメニューをカーソル位置に表示 

MButton::PopupMenuActivate(0, , "{MButton}") ; 選択語句を検索するポップアップメニューをマウス位置に表示 

!^MButton::KanzasiInfoGet() ; スクリプト修正用の情報を取得して表示 

^+n::NextDictActivate(1) ; 次の辞書をアクティブ化 

^+p::NextDictActivate(-1) ; 前の辞書をアクティブ化 

どのホットキーに対して何をするか、それぞれ 1行ずつで記述した形となっています。 

キーを変更するには、上記の青色で示した部分を、目的のホットキーを表す文字列に変更してください。キー

の指定方法は AHKの書式どおりで、「ホットキー設定ヘルパー」を使うと簡単です。 

※上の設定例で、3行目の「MButton」のみ、行頭と行中の 2カ所に出てきます。この機能を別のマウスボタンに変え

る場合は、同じように、行頭と行中の 2 カ所に記述してください。 

また、短縮入力のホットストリングをここで定義してもかまいません（その場合、かんざしの入力ウィンドウ以外

も含め、デスクトップ全体で有効な短縮入力となります）。指定方法の詳細については、AHKのヘルプなどを参

照してください。 

  

https://www.autohotkey.com/docs/v1/Hotstrings.htm

入力ウィンドウ上のショートカットキーの設定 

入力ウィンドウ上でのみ有効なショートカットキーを設定するセクションです。 

#IfWinActive かんざし - ahk_class AutoHotkeyGUI  

 

^2::                                                        ; 全体の " " 囲みのオン/オフの切り替え 

    hk_text := InputBoxGetText() 

    hk_text := InStr(hk_text, """") ? StrRepl(hk_text, """") : """" . hk_text . """" 

    InputBoxSetText(hk_text, 2) 

    return 

 

^3::                                                        ; 各単語の " " 囲みのオン/オフの切り替え 

    hk_text := InputBoxGetText() 

    hk_text := InStr(hk_text, """") ? StrRepl(hk_text, """") : RegExReplace(hk_text, "(*UCP)(¥S+)", """$1""") 

    InputBoxSetText(hk_text, 2) 

    return 

 

#IfWinActive 

デフォルトで設定されている「Ctrl+2」「Ctrl+3」というキーを変更したい場合は、上記の青色の部分を修正してく

ださい。 

また、短縮入力のホットストリングをここで定義してもかまいません。指定方法の詳細については、AHKのヘル

プなどを参照してください。たとえば、「,,a」「,,m」「,,g」という 3 種類の短縮入力を指定する場合、次のような 3

行を記述します。 

:*:,,a::site:apple.com 

:*:,,m::site:microsoft.com 

:*:,,g::site:google.com 

  

https://www.autohotkey.com/docs/v1/Hotstrings.htm
https://www.autohotkey.com/docs/v1/Hotstrings.htm


 

26 

 

4 スクリプト修正のヘルパー機能 

スクリプトの修正作業全般で利用する 2つのヘルパー機能について説明します。 

ホットキー設定ヘルパー 

スクリプトの中には、ホットキーやショートカットキーを指定する箇所がいくつかあります。こうした所では、たと

えば Ctrl+V は「^v」、Alt+F1 は「!F1」というように、AHK で定められた書式に従って、キーの組み合わせを表

す文字列を記述する必要があります。 

AHK版かんざしに同梱の「ホットキー設定ヘルパー」を使うと、キーの記述方法を簡単に確認できます。 

使い方は簡単です。 

(1) エクスプローラーで、Kanzasi.ahk と同じフォルダにある「ホットキー設定ヘルパー.ahk」をダブルクリック

してください。上のような画面が表示されます。 

(2) 「目的のホットキー」の欄で、指定したいホットキーを押してください。 

下の欄に、そのキーを表すコードが表示されます。 

スクリプトの中でキーを指定する部分には、このコードを貼り付けてください。 

ただしこのツールでは、一部の特殊なキーや、マウスのボタンを使うホットキーのコードは確認できません。

AHKのヘルプでご確認ください。 

※このツールで表示されるコードでは、a～zの大文字・小文字が、Kanzasi.ahk のデフォルトの記述と異なる場合が

ありますが、通常はどちらでも問題ありません。 

「Ctrl+Alt+中ボタン」 

スクリプトを修正するときには、「Ctrl+Alt+中ボタン」で表示されたコードを貼り付けるという形で対応する部分

がいくつかあります。 

この「Ctrl+Alt+中ボタン」には、次のような機能があります。 

入力ウィンドウ上での「Ctrl+Alt+中ボタン」 

入力ウィンドウ上で「Ctrl+Alt+中ボタン」をクリックすると、ウィンドウの x / y座標と幅の情報が表示されます。 

「Ctrl+Alt+中ボタン」の後のドロップダウンリスト 

アプリケーションのウィンドウや入力エリアの上で「Ctrl+Alt+中ボタン」すると、ローカル辞書ソフトとして検索す

るためのコードが表示されます。これは冒頭のローカル辞書登録で行ったとおりです。 

ここで、入力ウィンドウの右端の［▼］ボタンをクリックすると、ドロップダウン部分に、別のコードやデータがいく

つか入っています。スクリプトの中では、こうしたコードやデータを貼り付けて使う箇所もあります。 

リスト内の項目をクリックすると入力エリアに表示されますので、それをコピーして貼り付けてください。 

ただし、行頭にある 〈 〉 囲みの項目名はスクリプトでは不要ですので、その部分を取り除いた形で貼り付けて

ください。 

ドロップダウンリストに入っている項目は次のとおりです。 

項目 意味 

辞書ソフト検索用コード 
ローカル辞書ソフトに検索語句を送信するコード 

（入力エリアに表示されるコードと同一） 

非 URL型Web検索用コード URLに検索語句が含まれないタイプのサイトを検索するためのコード 

EXE自動起動用コード 自動起動させる辞書ソフトやブラウザを設定するためのコード 

ウィンドウタイトル マウス位置のウィンドウのタイトル部の内容 

ウィンドウクラス マウス位置のウィンドウのクラス* 

マウス位置相対座標 ウィンドウの左上を基点とした、マウスの位置 

マウス位置クラス マウス位置のコントロールのクラス* 

EXE フルパス マウス位置のウィンドウの EXE ファイルのフルパス 

* 「クラス」とは、プログラムの中でウィンドウやコントロールを指定するための識別子の一種です。 


